

## Application Note—AN0905A

---

Revision A

### **KMD-7003 and KMD-7053 VAV Controller**

This application note describes using a KMD-7003 and KMD-7053 for single-duct VAV installations. It covers using the controllers for heat/cool change over and proportional reheat.

---

#### **Related materials**

In addition to the material presented in this application note, review and have available the following reference materials.

- ◆ Installation and operation guides for KMD-7003 and STE-5013
- ◆ WinControl XL Plus User's Manual
- ◆ Hardware Configuration Manager reference guide
- ◆ System plans with controller addresses
- ◆ AN0805A, The NetSensor Balancer Program

#### **Introduction**

The KMD-7003 is a KMC digital controller and an actuator built into one unit. A pressure-independent factory programmed VAV sequence automatically modulates flow into the conditioned space between Minimum and Maximum setpoints as defined by the user. This sequence allows for Heating, Cooling, or changeover applications with minimal user initialization. Additionally, user programming can customize each application as required.

If you do not want to use any of the factory setup and programming options, send a panel file with your own setup parameters and programming to the controller. This will remove all of the factory configurations allowing you to program any desired sequence.

The KMD-7003 VAV Terminal Controller contains the basic programming sequence required for two specific applications:

- ◆ Single Duct VAV (heating-cooling-changeover)
- ◆ Single Duct VAV with time proportional reheat

In addition, there is a balance program for setting up VAV box parameters. Initially these devices will default to a standard Single Duct Cooling VAV configuration. You can use either of the Reheat functions by turning "ON" the respective programs and enabling a variable on the Variables screen. The following paragraphs provide specific details and explanations to help you use the KMD-7003.

## **Velocity sensor**

The KMD-7003 incorporates an on-board flow-through type sensor to determine duct airflow. Airflow through the sensor is approximately 1.12 SCFH (32 liters/hour) at .25 INWC (62.5 Pa) pressure drop. The sensor uses two RTD's (resistance temperature detectors) to measure flow. The RTD's are platinum films on a ceramic base chosen for long term stability and accuracy. The smaller 100 ohm RTD is self-heated to a constant temperature differential above ambient. The amount of power required varies with the specific airflow. A larger 1000-ohm RTD constantly senses ambient temperature and compensates. The sensor housing channels the airflow over the self-heated RTD in a manner to isolate its heat from the compensator. This minimizes shifts at low flows due to various mounting positions. The sensor housing itself is molded from plastic that carries the UL94-5V rating. The sensor ports are 0.188 inch diameter and readily accepts 0.25 inch O.D. "FR" tubing (.040 wall) or equivalent.

## **Velocity limits**

Minimum and Maximum Flow velocity limits can be defined in either of two ways: Through KMC digital operating software or with the NetSensor. Setting velocity limits through the KMC digital software is as simple as defining the value of a variable. Using this method the limits may be entered in any units desired (CFM, L/s, Pa, etc.) as long as a "conversion factor" is also provided. The "default" settings are always in Feet Per Minute (FPM). For details on using this method refer to the following section on standard VAV operation.

## **Damper actuator**

The actuator is capable of supplying a minimum of 50 inch pounds of torque to open or close the damper according to instructions from the KMD control sequence. The motor is an AC synchronous type for smooth, reliable operation and is coupled with a magnetic slip-clutch design that allows continuously stalling damper rotation without damage. A time-out function will stop the motor anytime the unit has run continuously in one direction for longer than 6 minutes. This ensures that the stops have been reached without needlessly running the motor.

Physical adjustable end-stops allow limiting rotation to less than the 90° rotation and a gear disengagement mechanism on the side of the case can be used for fast and simple installation and set-up. The red disengagement button is located on the left side of the actuator.

## **Standard Variable Air Volume (VAV) Operation**

The standard VAV sequence of operation has been pre-programmed into each KMD-7003 controller. Default operation is defined as:

- ◆ Single Duct Cooling
- ◆ Clockwise to close damper rotation
- ◆ Minimum and Maximum Limits at 500 and 2000 FPM respectively
- ◆ Cooling setpoint at 74°F
- ◆ Space Temperature Sensor connected to IN1

These units are capable of independent, stand-alone operation or they may be networked together with other KMC digital products. If stand-alone operation is desired, supplying 24 VAC power and connecting a 10K thermistor (STE-series) to

IN1 will enable full operation to the default parameters listed above. If these units will be networked together it may be necessary to configure each controller with a unique address and other set-up options using HCM.

The following is a simplified explanation of the control logic used in the standard VAV application:

- ◆ A full-function PID controller (CON1) is set to perform the thermostat function by comparing the space temperature (IN1) to the set point temperature (VAR25). In the COOL (default) changeover mode, thermostat action is Direct Acting and "Active" Setpoint is the Cooling setpoint temperature VAR30). Conversely, if HEAT changeover mode was selected, thermostat action would change to Reverse Acting and the "Active" set point would be the Heating Set Point (VAR29).
- ◆ A "requested" flow set point is calculated based on the thermostat information and Minimum / Maximum flow limits. Using this "Requested" flow a second full function PID Controller (CON2) looks at actual flow (IN4) as measured by the velocity sensor and instructs the damper to open or close accordingly.
- ◆ Parameters of either the Thermostat (CON1) or the Flow Controller (CON2) such as Proportional Band, Reset (Integral), or Rate (Derivative) functions can easily be changed to suit application requirements. The "BIAS" of 50% preset on CON2 should not be changed as this defines a satisfied "Neutral" position for the actuator drive circuit.

## Hardware Configuration Manager (HCM)

For networked operation, use the HCM software to configure the address, baud, and last panel setting. Each controller requires a distinct panel address and a common baud rate to reside on a network. The unit with the highest address must be designated as LAST PANEL. Before running the HCM software to set these parameters, flip the network isolation switch located near the terminal strip. See the Operators Manual for details on HCM and networking controllers.

**Table 11 Default Configuration for KMD-7003**

Panel Number	124
Network Baud Rate	38400
Last Panel	No
Panel Type	116

## Setting up with WinControl XL

This section covers the set-up requirements to activate any of the pre-programmed options available in this device. At this point, communications should have been established with the unit.

A "Master" password is available to "sign-on" to each controller. Other security codes can be entered to control access to the system.

If you do not want to use any of the factory setup and programming options, send a panel file with your own setup parameters and programming to the controller. This will remove all of the factory programming allowing you to program any desired sequence.

The first step after signing on to the controller is to “Load Descriptors One” or “Load Descriptors All”. Many of the descriptions have been pre-defined for Inputs, Outputs, and Variables to aid in configuring the options.

To use either the Time-Proportional Reheat (OUT1) program or the 3-stage Reheat (OUTS 1, 2, 3) program, go to the “Control Basic” programs screen and turn “ON” the appropriate program.

After the appropriate programs are enabled you can use the setup screen in Control Group #2 (GRP2) which lists the items most likely to be monitored. The following section outlines each of the parameters that have been pre-programmed into the device.

### Configuring inputs

Inputs 1, 2, and 3 can be selected through software for either analog or digital ranges. The pre-programmed sequences assume the following input configuration:

#	Description	Manual	Value	Units	Decom.	Label
1	ROOM TEMPERATURE	<input type="checkbox"/>	72.00	Deg. F	<input checked="" type="checkbox"/>	ROOMTEMP
2		<input type="checkbox"/>	0.00	Unused	<input type="checkbox"/>	
3		<input checked="" type="checkbox"/>	0.00	Unused	<input type="checkbox"/>	
4	AIR FLOW SENSOR	<input type="checkbox"/>	250.00	Unused (Table 2)	<input checked="" type="checkbox"/>	FLWSNSNR

Panel: A1      Status: Idle

**Illustration 7 Input configuration in WinControl XL**

**Input # 1–ROOM TEMPERATURE** Default selects a KMC 10K ohm thermistor as used in all of KMC’s STE series sensors.

**Input #2** Unused

**Input #3** Unused

**Input #4–Air Flow Sensor** The on-board flow-sensor. Table 2 has been configured to convert the voltage input into feet per minute. This input is not accessible for any other use.

Configuring outputs

The KMD-7003 VAV Controller features three user available outputs and an analog output to drive the damper motor. The KMC factory supplied control routines have been written to use the following output configuration:

#	Description	Manual	Value	Units	Label
1	STAGE 1	<input type="checkbox"/>	Off	Off/On	
2	STAGE 2	<input checked="" type="checkbox"/>	Off	Off/On	
3	STAGE 3	<input type="checkbox"/>	Off	Off/On	
4	DAMPER MOTOR DRIVE	<input type="checkbox"/>	0.00	Volts	MOTOR

End Auto Edit Erase OK Cancel

Panel: A1 Auto Load Mode [10]

Illustration 8 Output configuration in WinControl XL

**Output #1–Stage 1** The factory program defines Output 1 as the first stage of heating on VAV boxes with electric reheat (PRG3). Optionally, it can be used to control any 0-10 VDC device by adding your own custom programming in program area 5 (PRG5).

**Output #2–Stage 2** The factory program defines Output 2 as the second stage of heating on VAV boxes with electric reheat (PRG3). This is a triac that can be programmed to switch valve actuators or 24-volt reheat circuits. Optionally, it can be used to control any 24-volt circuit by adding your own custom programming in program area 5 (PRG5).

**Output #3–Stage 3** The factory program defines Output 3 as the third stage of heating on VAV boxes with electric reheat (PRG3). Optionally, it can be used to control any 24-volt circuit by adding your own custom programming in program area 5 (PRG5).

**Output #4–DAMPER MOTOR DRIVE** This output is used to control the damper position. It is not accessible for any other use on the KMD-7003. When the output is commanded to 0 volts the motor drives the damper counterclockwise, 10 volts drives the damper clockwise, and 5 volts stops the damper.

! Caution

When connecting loads to the triac (Output 2) or relay (Output 3), use only the terminal marked RET associated with the triac or relay for 24-volt circuits.

Configuring variables

A total of thirty-two system variables are available to perform various functions such as set points, limits, and modes. All variables can be changed or modified to suit the specific application. The following have been pre-defined for use in the KMD-7003 factory supplied control sequences:

**Variable #17–CHANGEOVER MODE (COOL/HEAT)** Chooses whether the unit will operate as a heating or cooling box by automatically setting the thermostat action as direct (COOL) or reverse (HEAT).

**Variable #18–CLOCKWISE TO CLOSE? (NO/YES)** Defines which way the damper will rotate to increase or decrease air-flow as required. “NO” will close the box in a counter-clockwise direction.

**Variable #19–MINIMUM FLOW** Sets the lowest design flow for the VAV box. The default setting is for a Feet-Per-Minute (FPM) value. Other units can be used by providing a conversion factor in Variable 21 (VAR21).

**Variable #20–MAXIMUM FLOW** Sets the highest design flow for the VAV box. The default setting is for a FPM value. Other units can be used by providing a conversion factor in Variable 21 (VAR21).

**Variable #21–VOLUME CONVERSION** Default value is 1.00 (no conversion). This factor allows all flow settings and live readings to be in any desired units. Since default calculations are in FPM, the duct area is required (in square feet) to get readings in CFM. Similarly, values can be converted to L/s, Pa, etc. with the proper factors applied.

**Variable #22–SENSOR CORRECTION** Default value is 1.00 (no correction required). This factor is provided to allow simple correction for erroneous sensor readings. This factor is multiplied against all flow readings and the product is stored as the ACTUAL FLOW (VAR23).

**Variable #23–ACTUAL FLOW** This value is calculated from the flow sensor input (IN4) and the sensor correction variable (VAR22) using the following formula:

$$\text{ACT-FLOW} = \text{FLOWSNSR} * \text{VAR22}$$

Or:

$$\text{VAR23} = \text{IN4} * \text{VAR22}$$

Units for this Variable must always be in Feet per Minute for proper operation of the built-in control sequences. For flow displayed in desired units see Variable 26.

**Variable #24–REQUESTED FLOW** This is the flow set point based on the heating or cooling needs of the space. The value is set automatically in Program 1 (PRG1). Units for this Variable must always be in Feet per Minute for proper operation of the built-in control sequences. For flow displayed in desired units see Variable 26.

**Variable #25–ACTIVE TEMP SETPOINT** This is the temperature the controller will attempt to maintain in the space. The active set point is the heating set point (VAR29) in HEAT MODE (VAR17 = ON) and the cooling set point (VAR30) in COOL MODE (VAR17 = OFF)

**Variable #26–DUCT VOLUME READOUT** This variable has been set up to allow easy access to a live velocity read-out. Units can be converted using the Volume Conversion factor (VAR21) to relate to CFM, L/s, Pa, etc. as needed. The following formula is used in Program 1 (PRG1) to obtain this value:

$$\text{VOLUME1} = \text{ACT-FLOW} * \text{VOLFACTR}$$

Or:

$$\text{VAR26} = \text{VAR23} * \text{VAR21}$$

**Variable #27–REHEAT STATUS** This flag will enable or disable the REHEAT function. If reheat is disabled, all outputs being used in that reheat function are overridden to “OFF”. To choose one of the pre-programmed Reheat options, the appropriate program must be turned on. See the Program Menu for options. If neither reheat program is turned on, this variable has no purpose.

**Note**

A common application is to use this flag to disable the Reheat function during the summer mode.

**Variable #28–AUXILIARY FLOW** Assigning a flow set point other than 0 in this variable allows the VAV set point to be overridden when STAGE 1 (OUT1) heat is turned “ON”. The auxiliary flow set point is only enabled if STAGE HEATING (PRG3) is used, the REHEAT STATUS (VAR27) is enabled, and the value is non-zero. To disable the auxiliary flow, leave the value at 0 (zero).

**Variable #29–HEAT TEMP SETPOINT** This is the active set point used by CON1 when in the heating mode (VAR17 = ON). Units must match that of the temperature sensor on IN1 (deg. F or deg. C)

**Variable #30–COOL TEMP SETPOINT** This is the active set point used by CON1 when in the cooling mode (VAR17 = OFF). Units must match that of the temperature sensor on IN1 (deg. F or deg. C)

**Variable #31–DAMPER DRIVE TIME** This is the approximate time in seconds required for the damper to move from the fully closed to the fully open position or from the fully open to the fully closed position. It is used in program 1 to roughly determine the damper position in variable 32 (VAR32).

Approximate travel time is 300 seconds for 90 degrees, 200 seconds for 60 degrees and 150 seconds for 45 degrees. The end stops on the back side of the KMD-7003 need to be placed in their appropriate places to get the desired rotation.

**Variable #32–DAMPER POSITION** This is a calculated position based on the full stroke drive time (VAR31). Variable 31 must be set correctly for the damper position information to work properly.

Damper position is determined in Program 1 by measuring the time the damper has been commanded open or closed versus the time needed for complete travel. If the damper is stuck this variable will not necessarily indicate the correct damper position. It will however indicate the controllers attempt to move the damper. This variable is used only to give users an indication of what the controller has done and is doing to control the position of the damper. It is not used as feedback in the control sequence.

**Configuring PID control loops**

KMC digital products feature built-in, full function PID loop controllers. All parameters may be modified as necessary.

**Controller #1 (CON1)** is a direct-acting thermostat. It compares ROOM TEMPERATURE (IN1) to the ACTIVE TEMP SETPT (VAR25) to determine the VAV flow requirement.

**Controller #2 (CON2)** is set up as a direct-acting thermostat. It compares ACTUAL FLOW (VAR23) to the REQUESTED FLOW (VAR24) to operate the DAMPER (OUT4). All values used by the controller have been converted to FPM to allow a common proportional band value regardless of the units selected. Default is a 200 FPM proportional band which relates to approximately 60 FPM “dead band” to stabilize actuator operation. Default parameters do not include any integral or derivative function.



---

#### Note

The desired air flow is VAR24 and is automatically calculated by the VAV program. For example, to override the airflow to some other value under specific conditions, include a statement such as:

```
10 IF (condition) THEN VAR24 = (value)
```

All values at this point must be in FPM. Condition may be a variable that represents an abnormal event such as a smoke zone activation. To ensure close-off on a box, set the desired flow (value) to a number below zero such as -50. To ensure a wide open damper position use a flow rate higher than 3000 FPM (The highest value the on-board sensor can read).

**Controller #3 (CON3)** compares the ROOM TEMPERATURE (IN1) to the HEATING SET POINT (VAR29) for operation of the reheat stages when programs 2 or 3 are enabled.

#### Control Basic programming

The first program handles set points and flow control. There are two programs provided to control reheat in the VAV box. If neither sequence fits the required application, edit the programs or send blank programs to the controller and replace any of the factory programs. The factory programs can be retrieved if necessary. The original factory programs can be re-installed again by using the Clear Panel function in the System Setup Menu.



---

#### Caution

This will erase any programming changes that have been made to the controller. (Clear Panel does not affect changes made with HCM, or panel type).


**Program 1–Single Duct VAV** This program sets the ACTIVESP (VAR25) based on the CHANGEOVER MODE (VAR17). When the CHANGEOVER MODE is OFF the ACTIVE SET POINT equals the Heating Set Point (VAR29). When the CHANGEOVER MODE is ON the ACTIVE SET POINT equals the Cooling Set Point (VAR30). The necessary Air flow is calculated and the damper motor is modulated to maintain this volume.

See [Program 1– SINGLE DUCT VAV](#) on page 3-61 for a complete listing of the program.

**Program Two–TIME PROPORTIONAL REHEAT** See [Program 2–TIME PROPORTIONAL HT](#) on page 3-62 for a complete listing. As supplied from the factory, this uses only one digital output (OUT1) by controlling the percentage of




"ON" time over some specific time interval. For the pre-programmed sequence a fixed 10-second time base is used; A call for 73% heating would turn "ON" the output for 7.3 seconds and turn "OFF" for the remaining 2.7 seconds. See Illustration 10. Using this method of reheat requires only one output. Outputs 2 and 3 are left for other output devices. Illustration 9 shows a graphical representation of Program 1.



**Illustration 9 Proportional Reheat**

**Program 3—Three-stage Reheat** As supplied from the factory, this program sequentially steps each stage of reheat on as the requirement for heating increases (using Outputs 1, 2, and 3). Each stage turns "ON" at a fixed point (30%, 60%, and 90%) of the heating thermostat (CON3) value. The controller changes 0-100% over the proportional band range. Illustration 10 shows the sequence graphically.



**Illustration 10 Three-stage reheat**

This program may also be used for FAN-BOX applications. Connect the FAN relay to STAGE 1 (OUT1) and 2 stages of reheat to outputs 2 and 3. A differential is built into the sequence to prevent cycling the outputs over small temperature variations.

An optional AUXILIARY FLOW limit has been pre-programmed for use with the 3-stage Reheat sequence. This feature allows increasing the VAV flow set point to some value other than Minimum Flow anytime Stage 1 of heating is enabled. The value of Auxiliary flow entered in VAR28 must be in the same units as Minimum

and Maximum flow since the same “conversion factor” (VAR21) is used by the programming. If this feature is NOT being used, leave the Auxiliary Flow (VAR28) value at zero (0) and the program will ignore it.

The programming has been written to instruct the unit to control flow at the Auxiliary Flow limit whenever conditions call for stage 1 heating. If it is desired to initiate the Auxiliary Flow before actually turning “ON” the first stage of heat, add “Delay” time in the appropriate column of OUT1 on the Outputs screen (up to 99 min. 50 sec.). The delay will take precedence over all other software actions.

See [Program 3 3–STAGE REHEAT](#) on page 3-62 for a complete listing

**Program 4–Balancer override** This program is used with the NetSensor to force the controller to maintain either a minimum or a maximum air flow setpoint. Details on using the NetSensor can be found in applications note AN0805A, *The NetSensor Balancer Program*.

### Program 5–Available for user programming

#### Configuring tables

Custom tables allow the user to create unique “ranges” to be used by inputs or other special applications. The most often used application of these custom tables is to establish a correlation between a

potentiometer setting (a voltage) and a temperature value (such as the dial setting on a thermostat). The KMD-7000 Series Controllers allow user-access to all 3 tables. These tables have been defined for the following applications in the KMD-7003:

**Table #1** Ready for user programming.

**Table #2** Correlates velocity pressure from the on-board flow sensor to feet per minute.

**Table #3** Correlates velocity pressure of CSP-5001 VAV controller to feet per minute.



---

#### Note

These tables use nominal values for correlation. Due to component tolerances and the resistance of field wiring, these values may need correction on an individual basis. If the value read is too low, enter a positive calibration in the input column labeled “cal”. If the value read is high, a negative calibration can be entered. If necessary, reconfigure Table #2 and/or Table #3. The complete Table #2 and Table #3 from the factory are shown in figure 3. If the tables have been reconfigured and the user wants the factory configuration for Table #2 and Table #3 it will be necessary to use Clear Panel or re-enter the values.

**Caution**

Using Clear Panel will erase any programming changes that have been made to the controller. (Clear Panel does not affect changes made with HCM, or panel type.

Tables					
#	Table 1	Unused	Table 2	Unused	Table 3
1	0	0	0	0	0
2	0	0	0.29	0	0.35
3	0	0	0.4	92	0.46
4	0	0	0.8	401	0.86
5	0	0	1	530	1.1
6	0	0	1.5	803	1.7
7	0	0	2	1032	2.3
8	0	0	2.5	1253	2.5
9	0	0	3	1505	2.9
10	0	0	3.5	1824	3.25

Panel: A1 Edit Mode

**Illustration 11 Table configuration in WinControl XL**

The primary control logic in this model has a heating and cooling setpoint defined in the software as variables. For applications requiring room set point control at the thermostat, follow these steps:

1. Define an Input as the base set point—such as IN2 using the STE-5000 Series Room thermostat. Enter Table 1 in the Range column of the Input window.

2. Set up the table as shown in Table 12 or Table 13. The WinControl Table 1 will convert the voltage from the dial into degrees Fahrenheit.

#	Description	Manual	Value	Units	Decom.	Label
1	SPACE TEMP	<input checked="" type="checkbox"/>	72.80	Deg. F	<input type="checkbox"/>	SPACETP
2	BASE SET POINT	<input type="checkbox"/>	72.0	Deg. F	<input type="checkbox"/>	SPACESP

Auto Load End Edit Erase OK Cancel

Panel: A1 Edit Mode

**Illustration 12** Alternate input configuration in WinControl XL

**Table 12** STE-5200 Conversion Table

	Value	
1	0.00	86.0
2	0.710	85.0
3	1.07	80.0
4	1.38	75.0
5	1.67	70.0
6	1.91	65.0
7	2.10	60.0
8	2.23	56.0

**Table 13** STE-5000 Conversion Table

	Value	
1	0.00	54.0
2	0.540	58.0
3	1.27	66.0
4	1.56	70.0
5	1.80	74.0
6	2.01	78.0
7	2.50	90.0
8	3.00	90.0

3. Add programming in program area #5 to set the Heat and Cool set points. The following lines will set the Heating set point to 2° below the set point shown on the thermostat dial and the Cooling set point to 2° above the thermostat dial setting.

*Example:*

```
10 VAR29 = IN2 - 2
20 VAR30 = IN2 + 2
```

## Adding fan control

The following topics modify Program 2 and Program 3 to include fan control.

### Modifications to Program 2

The following modification adds fan control to the supplied time-proportional reheat program.

- ◆ Change all references to *Stage 1* to *Stage 2*.
- ◆ Label Output 3 as FAN
- ◆ Replace Program 2 with the following Control Basic program:

*Example:*

```

10 REM ----- TIME PROPORTIONAL REHEAT SEQUENCE
-----
20 REM -- Designed to drive a solid state relay connected to
electric reheat.
30 REM Time base is fixed at 10 seconds - ie, a call for 50%
heat turns on
40 REM relay (OUT1) for 5 seconds, off for 5 seconds.
50 REM Uses CON#3 for control of reheat operation! ( Full PID
function )
60 IF NOT VAR27 THEN STOP OUT1 , END
70 K = 1 / SCANS
80 L = L + K : IF L > 10 THEN L = 0
90 IF L < CON3 / 10 THEN START OUT1 ELSE STOP OUT1
100 IF CON3 > 30 THEN START OUT3
110 IF CON3 < 10 THEN STOP OUT3
120 END

```

### Modifications to Program 3

The following modification changes the supplied three-stage reheat program to a two-stage reheat with fan control.

- ◆ Delete all references to *Stage 3*
- ◆ Label Output 3 as FAN
- ◆ In Program 3, starting at Line 60, replace all programs lines with the following lines of Control Basic:

*Example:*

```

60 IF NOT VAR27 THEN STOP OUT1 , STOP OUT2 , STOP OUT3 , STOP A
, END
70 IF CON3 > 15 THEN START OUT1 , START A
80 IF CON3 > 60 THEN START OUT2
90 IF CON3 < 40 THEN STOP OUT2
100 IF CON3 < 11 THEN STOP OUT1 , STOP A
110 IF A THEN VAR24 = VAR28 / VAR21
120 IF OUT1 OR OUT2 THEN START OUT3
130 IF NOT OUT1 AND NOT OUT2 THEN STOP OUT3
140 IF A THEN VAR24 = VAR28 / VAR21
150 END

```

## Using the NetSensor

A NetSensor can be used to bring values into the controller without using any physical input. To add a NetSensor and use the factory programming in the KMD-7003, change the PID controllers #1 and #3 in-label from IN1 to the variable mapped to the NetSensor's onboard temperature sensor.

## Setback (unoccupied) setpoints

Many applications require “setback” or “setup” temperatures as a means of energy savings in unoccupied periods. To accomplish this, it is necessary to establish a condition of occupancy and define when and how these modes occur.

A variable can be set up as a “flag” to indicate the current mode of operation. The units “Un/Occupy” work well for this purpose.

#	Description	Manual	Value	Units	Decom.	Label
1	SPACE TEMP	<input checked="" type="checkbox"/>	72.80	Deg. F	<input type="checkbox"/>	SPACETP
2	BASE SET POINT	<input type="checkbox"/>	72.0	Deg. F	<input type="checkbox"/>	SPACESP

Buttons: Auto Load, End Edit, Erase, OK, Cancel

Panel: A1 Edit Mode

**Illustration 13** Input option for occupancy setpoints

Program 5 can contain the code needed to change the state of this Variable based on a weekly schedule as shown in this example:

```
10 IF WS1 THEN START OCCUPIED ELSE STOP OCCUPIED
```

The flag “OCCUPIED” (VAR1 in this example) is “ON” in the occupied mode and off in the unoccupied mode. Two more variables can be set up to define the heating and / or cooling set-back temperatures.

#	Description	Manual	Value	Units	Label
1	OCCUPANCY STATUS	<input type="checkbox"/>	Unoccupied:Unocc/Occ		OCCUPIED
2	HEATING SETBACK	<input type="checkbox"/>	65.0	Deg. F	HEATSB
3	COOLING SETBACK	<input type="checkbox"/>	0.0	Deg. F	COOLSB

Buttons: Auto Load, Edit, Erase, OK, Cancel

Panel: A1 Status: Idle

**Illustration 14** Alternate variables

Now, some simple programming can be written to determine the setpoint based on the occupancy mode:

```
10 IF OCCUPIED THEN HEAT-SP = SPACESP - 2 , COOL-SP = SPACESP +  
2  
20 IF NOT OCCUPIED THEN HEAT-SP = HEATSB , COOL-SP = COOLSB
```

When in the occupied mode, the heating set point is 2 degrees less than the set point on the thermostat dial and the cooling set point is 2 degrees above. In the unoccupied mode (line 20), the heating set point is set to whatever value is in VAR2 and the cooling set point is set equal to VAR3. It is not necessary to use variables for heating set-back or cooling set-up, but doing so makes modification of these values much easier than getting into the program to change them.

## Proportional reheat

A short line of code can be entered into Program area 5 to control hot water reheat. Since Controller #3 (CON3) is set up as a reverse acting thermostat, its value will change from 0 to 100% as temperature falls below the heating setpoint. Configure the output to a range of 0-100% then use:

```
40 OUT1 = CON3
```

## Other options

One of the biggest features of any KMC digital DDC Controller is the inherent programming ease and flexibility available in each and every unit. If the KMC supplied sequences do not satisfy the application at hand they may be turned off or edited and the entire control strategy can be manipulated. The WinControl XL User's Manual provides help and examples for programming Control Basic.

**WinControl settings** The following tables display the WinControl XL Plus settings for this application note.

**Table 14 Input configuration**

	Description	Manual	Value	Units	Average	Label
1	ROOM TEMPERATURE	auto	74.0 deg.F	K10K-40->250	64	ROOMTEMP
2		auto	0.00	Unused		
3		auto	0.00	Unused		
4	AIR FLOW SENSOR	auto	1115	Table 2	64	FLOWSNSR

**Table 15 Output configuration**

	Description	Manual	Output	Units	0 %	100	Delay	Security	Label
1	STAGE 1	auto	Off	Off / On			0:00	0	
2	STAGE 2	auto	Off	Off / On			0:00	0	
3	STAGE 3	auto	Off	Off / On			0:00	0	
4	DAMPER MOTOR DRIVE	auto	5.00 Volts	0.0 -> 10	0.0	10		0	MOTOR

**Table 16 Variables**

	Description	Manual	Value	Units	Label
1-16	(Available for user programs)				
17	CHANGEOVER MODE	auto	auto	Cool	Cool/Heat
18	CLOCKWISE TO CLOSE ?	auto	auto	Yes	No / Yes
19	MINIMUM FLOW	auto	auto	500.00	MIN-FLOW
20	MAXIMUM FLOW	auto	auto	2000.00	MAX-FLOW
21	VOLUME CONVERSION	auto	auto	1.00	VOLFACTR
22	SENSOR CORRECTION	auto	auto	1.00	
23	ACTUAL FLOW	auto	auto	0.00	ACT-FLOW
24	REQUESTED FLOW	auto	auto	0.00	RQSTFLOW
25	ACTIVE TEMP SETPOINT	auto	auto	74.00	ACTIVESP
26	DUCT VOLUME READOUT	auto	0.00		VOLUME1
27	REHEAT STATUS	auto	Disabled	Dis/Enable	REHEAT
28	AUXILLARY FLOW	auto	auto		AUX-FLOW
29	HEAT TEMP SETPOINT	auto	68.00	deg.F	HEAT-SP
30	COOL TEMP SETPOINT	auto	74.00	deg.F	COOL-SP
31	DAMPER DRIVE TIME	auto	300.00	Sec.	DRV-TIME
32	DAMPER POSITION	auto	50.00	%	DMPR-POS



**Table 17 Controllers**

	In Label	In Value	A/M	Output	Setpoint	Value	Action	Prop	Reset	Rate	Bias	
1	ROOMTEMP	0.00 deg.F	auto	50.0 %	ACTIVESP	74.0	+	2	0	H	0.00	50
2	ACT-FLOW	0.00	auto	0.0 %	RQSTFLOW	1250	+	200	0	H	0.00	50
3	ROOMTEMP	0.00 deg.F	auto	0.0 %	HEAT-SP	68.0	-	2	0	H	0.00	0
4		0.00	auto	0.0 %		0.00	+	0	0	H	0.00	0

**Table 18 Tables**

	TABLE 1 Value	TABLE 2 Value	TABLE 3 Value
1	0.00	0.00	0.00
2	0.00	0.00	0.140
3	0.00	0.00	0.200
4	0.00	0.00	0.400
5	0.00	0.00	0.800
6	0.00	0.00	1.00
7	0.00	0.00	1.50
8	0.00	0.00	2.00
9	0.00	0.00	2.50
10	0.00	0.00	3.00
11	0.00	0.00	3.50
12	0.00	0.00	4.00
13	0.00	0.00	4.50
14	0.00	0.00	5.00
15	0.00	0.00	5.50

**Table 19 NET SENSOR #1**

Button	Point	Value	Type	Range	Display Only
1		0.1	Analog	Deg.F	
2	ACTIVE TEMP SETPOINT	0	Analog	0 Decimal	Can Change
3	DUCT VOLUME READOUT	0	Analog	0 Decimal	Display Only
4	VOLUME CONVERSION	0.00	Analog	2 Decimal	Can Change
5	MINIMUM FLOW	0	Analog	0 Decimal	Can Change
6	SENSOR CORRECTION	0.00	Analog	2 Decimal	Can Change
7	MAXIMUM FLOW	0	Analog	0 Decimal	Can Change
Aux	BALANCER override	Off	State	Off/On	

***Illustration 15 Control Basic program list***

	<b>Description</b>	<b>Run</b>	<b>A/M</b>	<b>Exit</b>	<b>Label</b>
1	SINGLE DUCT VAV	Yes	auto	No	
2	TIME PROPORTIONAL HT	No	auto	No	
3	3-STAGE REHEAT	No	auto	No	
4	BALANCER override	No	auto	No	
5	USER PROGRAMS	No	auto	No	

**Program 1– SINGLE DUCT VAV**

```

1 REM BALANCER OVRD
5 IF TIME-ON( BALANCER~override ) > 0:30:00 THEN STOP
  BALANCER~override
7 IF BALANCER~override THEN END

10 IF CHANGEOVER~MODE THEN ACTIVE~TEMP~SETPOINT =
  HEAT~TEMP~SETPOINT ELSE ACTIVE~TEMP~SETPOINT =
  COOL~TEMP~SETPOINT
20 IF CHANGEOVER~MODE THEN A = 1 - CON1 / 100 ELSE A = CON1 /
  100
30 REQUESTED~FLOW = ( ( MAXIMUM~FLOW - MINIMUM~FLOW~ ) * A +
  MINIMUM~FLOW~ ) / VOLUME~CONVERSION
40 ACTUAL~FLOW~ = AIR~FLOW~SENSOR * SENSOR~CORRECTION : REM
  ACTUAL FLOW CORRECTED
50 DUCT~VOLUME~READOUT = ACTUAL~FLOW~ * VOLUME~CONVERSION : REM
  LIVE VOLUME

60 IF CON2 > 70 AND CLOCKWISE~TO~CLOSE~? THEN START B
70 IF CON2 > 70 AND NOT CLOCKWISE~TO~CLOSE~? THEN START C
80 IF CON2 < 30 AND CLOCKWISE~TO~CLOSE~? THEN START C
90 IF CON2 < 30 AND NOT CLOCKWISE~TO~CLOSE~? THEN START B
100 IF CON2 > 35 AND CON2 < 65 THEN STOP B , STOP C
110 IF+ B THEN STOP C
120 IF+ C THEN STOP B

125 X = 1 / SCANS
130 IF CON2 < 99 AND CON2 > 1 THEN 140 ELSE 210
140 IF CON2 < 36 OR CON2 > 64 THEN START D ELSE STOP D
150 IF TIME-ON( D ) > 0:04:00 THEN 210
170 Y = X + Y
180 IF Y > 5 THEN Y = 0
190 Z = ABS( CON2 - 50 ) / 50
200 IF Y > Z THEN STOP B , STOP C

210 IF B THEN DAMPER~MOTOR~DRIVE = 10
220 IF C THEN DAMPER~MOTOR~DRIVE = 0
230 IF NOT B AND NOT C THEN DAMPER~MOTOR~DRIVE = 5
234 IF DAMPER~MOTOR~DRIVE > 7 THEN START G ELSE STOP G
236 IF DAMPER~MOTOR~DRIVE < 3 THEN START H ELSE STOP H
240 IF TIME-ON( G ) > 0:06:00 OR TIME-ON( H ) > 0:06:00 THEN
  DAMPER~MOTOR~DRIVE = 5 , START I ELSE STOP I
246 IF TIME-ON( I ) > 6:00:00 THEN STOP G , STOP H , STOP B ,
  STOP C

250 IF CLOCKWISE~TO~CLOSE~? AND DAMPER~MOTOR~DRIVE < 3 THEN
  DAMPER~POSITION = DAMPER~POSITION + X * 100 / DAMPER~DRIVE~TIME
260 IF CLOCKWISE~TO~CLOSE~? AND DAMPER~MOTOR~DRIVE > 7 THEN
  DAMPER~POSITION = DAMPER~POSITION - X * 100 / DAMPER~DRIVE~TIME
270 IF NOT CLOCKWISE~TO~CLOSE~? AND DAMPER~MOTOR~DRIVE < 3 THEN
  DAMPER~POSITION = DAMPER~POSITION - X * 100 / DAMPER~DRIVE~TIME
280 IF NOT CLOCKWISE~TO~CLOSE~? AND DAMPER~MOTOR~DRIVE > 7 THEN
  DAMPER~POSITION = DAMPER~POSITION + X * 100 / DAMPER~DRIVE~TIME

290 IF DAMPER~POSITION > 100 THEN DAMPER~POSITION = 100
300 IF DAMPER~POSITION < 0 THEN DAMPER~POSITION = 0
310 END

```

### Program 2–TIME PROPORTIONAL HT

The following program is supplied with models KMD-7003 and KMD-7053. To modify the program for time-proportional reheat with a fan, see [Adding fan control](#) on page 3-55 for modification instructions.

```
20 REM ----- TIME PROPORTIONAL REHEAT SEQUENCE
-----
25 REM -- Designed to drive a solid state relay connected to
electric reheat.
26 REM Time base is fixed at 10 seconds - ie, a call for 50%
heat turns on
27 REM relay (OUT1) for 5 seconds, off for 5 seconds.
28 REM Uses CON#3 for control of reheat operation! ( Full PID
function )
30 IF NOT REHEAT~STATUS THEN STOP STAGE~1 , END
40 K = 1 / SCANS
50 L = L + K : IF L > 10 THEN L = 0
60 IF L < CON3 / 10 THEN START STAGE~1 ELSE STOP STAGE~1
70 END
```

### Program 3 3–STAGE REHEAT

The following program is supplied with models KMD-7003 and KMD-7053. To modify the program for 2-stage reheat with a fan, see [Adding fan control](#) on page 3-55 for modification instructions.

```
10 REM ----- 3-STAGE REHEAT SEQUENCE
-----
20 REM -- STAGES ON AT 30, 60, 90 % OF PROPORTIONAL BAND BELOW
SETPOINT.
30 REM Auxillary flow setpoint is used only when AUX-FLOW
(VAR28) is not
40 REM zero and anytime stage 1 heat (OUT1) is requested.
50 REM Entire reheat sequence is easily disabled by toggling
VAR27 on-off.
60 IF NOT REHEAT~STATUS THEN STOP STAGE~1 , STOP STAGE~2 , STOP
STAGE~3 , STOP A , END
70 IF CON3 > 30 THEN START STAGE~1 , START A
80 IF CON3 > 60 THEN START STAGE~2
90 IF CON3 > 90 THEN START STAGE~3
100 IF CON3 < 80 THEN STOP STAGE~3
110 IF CON3 < 50 THEN STOP STAGE~2
120 IF CON3 < 20 THEN STOP STAGE~1 , STOP A
130 IF AUXILLARY~FLOW = 0 THEN 150 : REM Since Auxillary flow is
zero its not used
140 IF A THEN REQUESTED~FLOW = AUXILLARY~FLOW /
VOLUME~CONVERSION
150 END
```

**Program 4–BALANCER override**

```

10 REM **** Balancer control of unit to either MIN or MAX flow
   ***
20 REM ** Uses var25 activesp in digital fashion to select
   desired flow **

25 IF+ ACTIVE~TEMP~SETPOINT > 1 THEN ACTIVE~TEMP~SETPOINT = 1
27 IF+ ACTIVE~TEMP~SETPOINT < 0 THEN ACTIVE~TEMP~SETPOINT = 0
30 IF ACTIVE~TEMP~SETPOINT THEN REQUESTED~FLOW = MAXIMUM~FLOW /
   VOLUME~CONVERSION ELSE REQUESTED~FLOW = MINIMUM~FLOW~ /
   VOLUME~CONVERSION
40 ACTUAL~FLOW~ = AIR~FLOW~SENSOR * SENSOR~CORRECTION : REM
   ACTUAL FLOW CORRECTED (FPM)
50 DUCT~VOLUME~READOUT = ACTUAL~FLOW~ * VOLUME~CONVERSION : REM
   LIVE VOLUME READOUT - UNITS MATCH VOLUME

60 IF CON2 > 70 AND CLOCKWISE~TO~CLOSE~? THEN START B
70 IF CON2 > 70 AND NOT CLOCKWISE~TO~CLOSE~? THEN START C
80 IF CON2 < 30 AND CLOCKWISE~TO~CLOSE~? THEN START C
90 IF CON2 < 30 AND NOT CLOCKWISE~TO~CLOSE~? THEN START B
100 IF CON2 > 35 AND CON2 < 65 THEN STOP B , STOP C
110 IF+ B THEN STOP C
120 IF+ C THEN STOP B

125 X = 1 / SCANS
130 IF CON2 < 99 AND CON2 > 1 THEN 140 ELSE 210
140 IF CON2 < 36 OR CON2 > 64 THEN START D ELSE STOP D
150 IF TIME-ON( D ) > 0:04:00 THEN 210
170 Y = X + Y
180 IF Y > 5 THEN Y = 0
190 Z = ABS( CON2 - 50 ) / 50
200 IF Y > Z THEN STOP B , STOP C

210 IF B THEN DAMPER~MOTOR~DRIVE = 10
220 IF C THEN DAMPER~MOTOR~DRIVE = 0
230 IF NOT B AND NOT C THEN DAMPER~MOTOR~DRIVE = 5

250 IF CLOCKWISE~TO~CLOSE~? AND DAMPER~MOTOR~DRIVE < 3 THEN
   DAMPER~POSITION = DAMPER~POSITION + X * 100 / DAMPER~DRIVE~TIME
260 IF CLOCKWISE~TO~CLOSE~? AND DAMPER~MOTOR~DRIVE > 7 THEN
   DAMPER~POSITION = DAMPER~POSITION - X * 100 / DAMPER~DRIVE~TIME
270 IF NOT CLOCKWISE~TO~CLOSE~? AND DAMPER~MOTOR~DRIVE < 3 THEN
   DAMPER~POSITION = DAMPER~POSITION - X * 100 / DAMPER~DRIVE~TIME
280 IF NOT CLOCKWISE~TO~CLOSE~? AND DAMPER~MOTOR~DRIVE > 7 THEN
   DAMPER~POSITION = DAMPER~POSITION + X * 100 / DAMPER~DRIVE~TIME

290 IF DAMPER~POSITION > 100 THEN DAMPER~POSITION = 100
300 IF DAMPER~POSITION < 0 THEN DAMPER~POSITION = 0
310 END

```

Application drawings




Illustration 16 Variable Air Volume with hot-water reheat



